# Maintainer: Sébastien Luttringer <seblu@archlinux.org>
# Contributor: Andreas Hauser <andy-aur@splashground.de>

pkgname=nload
pkgver=0.7.4
pkgrel=7
pkgdesc='Monitors network traffic and bandwidth usage'
url='http://www.roland-riegel.de/nload'
license=('GPL2')
depends=('gcc-libs' 'ncurses')
arch=('loong64' 'x86_64')
source=("$url/$pkgname-$pkgver.tar.gz")
sha512sums=('ca3e36d3f77c5a7994a3261042fb65a1458a29c769e6bebbe3b8e49b3045aca6f4d047065d11431b28368e0d881c24c92533fa24f0e54ac270f1f5bccfb5c0af')

prepare() {
  cd $pkgname-$pkgver
  # make gzip reproducible by not embedding timestamps
  sed -i 's/gzip -f/gzip -nf'/ docs/Makefile.am docs/Makefile.in
}

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
